
Après tous les exemples réalisés dans une multitude de langages, en voila en COBOL. 

J'ai utilisé GnuCOBOL, car j'ai pas de z/OS ou d'AS400 sous la main. Et j'ai pu utilisé la libc pour le gettimeofday. De plus GnuCOBOL ajoute quelque spécificités, en particulier sur le verbe ACCEPT.


             ******************************************************************
            * author/copyright goernil
            * date: 20/03/2018
            * this program is free software. it comes without any warranty, to
            * the extent permitted by applicable law. you can redistribute it
            * and/or modify it under the terms of the do what the fuck you
            * want to public license, version 2, as published by sam hocevar.
            * see http://www.wtfpl.net/ for more details.
            *
            * purpose: calculate tempo
            * tectonics: cobc
            ******************************************************************
            *
            identification division.
            program-id. taptempo.
            data division.
            file section.
            working-storage section.
                01 ws-timeval-64bit.
                    02 ws-seconds       pic x(8) comp-5.
                    02 ws-microseconds  pic x(8) comp-5.
                77 ws-tzp usage pointer value null.
                77 ws-ms-since-epoch    pic 9(18) display.

                01 ws-entree-clavier    pic x(80).

                01 ws-stockage-tap.
                    05 ws-tap pic 9(18) display occurs 100.
                01 ws-precision         pic 9(20) value 6.
                01 ws-precision-disp    pic z9.
                01 ws-i                 pic 9(3).
                01 ws-j                 pic 9(3).
                01 ws-tempo             pic z(8)9.
                01 ws-tap-moy-s         pic 9(9)v9(9).
                01 ws-nb-args           pic 9(9).
                01 ws-args.
                    03 ws-tab-arg occurs 2048.
                    05 ws-arg               pic x(20). 
                    05 ws-arg-num redefines ws-arg pic 9(20). 
                01 ws-max-nb-args       pic 9(9) values 2048.
                01 ws-arg-i             pic 9(9).

            procedure division.
            main-procedure.

                perform lecture-arguments.

                display 'Appuyer sur la touche entrée en cadence. '
                        'Le premier ne compte pas.'
                            '(q pour quitter)'
                accept ws-entree-clavier

                move 0  to ws-i

                perform until   ws-entree-clavier = 'q'
                    accept ws-entree-clavier
                    perform gettimeofday
                    if ws-i < ws-precision  
                    then
                        add 1 to ws-i
                        move ws-ms-since-epoch to ws-tap (ws-i)
                    else
                        perform pushtap
                    end-if

                    if ws-i = ws-precision  
                        compute ws-tap-moy-s =  ( ws-tap (ws-i) - ws-tap(1) )
                                        /  (ws-precision - 1)
                                        / 1000
                        compute ws-tempo = 1 / ws-tap-moy-s * 60
                        if ws-entree-clavier <> 'q'
                            display 'tempo : ' ws-tempo ':'
                        end-if
                        end-if
                end-perform

                perform endprogram           
                stop run.

            lecture-arguments.
                accept ws-nb-args from argument-number

                Perform until ws-arg-i = ws-nb-args
                    add 1 to ws-arg-i
                    accept ws-arg(ws-arg-i) FROM ARGUMENT-VALUE
                    evaluate  ws-arg(ws-arg-i)
                        when '-h'
                            perform display-help
                            perform endprogram
                        when '-s'
                            accept ws-precision  FROM ARGUMENT-VALUE
                            add 1 to ws-arg-i
                            move   ws-precision  to ws-arg(ws-arg-i)
                            if ws-precision <> 0
                                and ws-precision > 1
                                and ws-precision < 100
                            then 
                                move ws-precision to ws-precision-disp 
                                display 'la précision est ' 
                                        ws-precision-disp 
                                ' échantillons(s)'
                            else
                                display '-s usage : Doit être numerique '
                                    'et entre 2 et 99. Valeur par défault 6.'
                                perform endprogram    
                            end-if
                        when other
                            display 'Wrong args'
                            perform display-help
                            perform endprogram
                    end-evaluate
                end-perform
                continue.

            display-help.
                display 'Usage: taptempo '
                display '[-s samples]'
                continue.

            gettimeofday.
                call "gettimeofday" using
                    by reference ws-timeval-64bit
                    by value ws-tzp
                end-call
                compute ws-ms-since-epoch = ws-seconds * 1000
                        + ws-microseconds / 1000
                continue.

            pushtap.
                perform varying ws-j from 1 by 1 until ws-j = ws-i
                    move ws-tap (ws-j + 1) to ws-tap (ws-j)
                end-perform
                move ws-ms-since-epoch to ws-tap (ws-i)
                continue.

            endprogram.
                stop run
                continue.

            end program taptempo.

